United States
Department of
Agriculture

Forest Service

**Pacific Southwest
Forest and Range
Experiment Station**

General Technical
Report PSW-78

DOSESCREEN:
a computer program to aid dose placement

Kimberly C. Smith Jacqueline L. Robertson



The Authors:

are with the Station's unit developing improved technology for integrated management of western forest insects, with headquarters in Berkeley, Calif. **KIMBERLY C. SMITH,** a mathematical statistician, earned a bachelor's degree in mathematics (1975) at the University of Florida and a master's degree in statistics (1979) at the University of New Mexico. She joined the Station's staff in 1980. **JACQUELINE L. ROBERTSON,** a research entomologist, is in charge of the unit. She earned a bachelor's degree in zoology (1969) and a doctorate in entomology (1973) at the University of California, Berkeley. She has been a member of the Station's research staff since 1966.

Publisher:

Pacific Southwest Forest and Range Experiment Station
P.O. Box 245, Berkeley, California 94701

August 1984

DOSESCREEN: a computer program to aid dose placement

Kimberly C. Smith Jacqueline L. Robertson

CONTENTS

The most common method used to evaluate the response of insects to chemicals, such as insecticides, is the dose-response bioassay. An important aspect of designing a chemical bioassay is dose placement--selecting doses of the chemical for testing to obtain effective dose (ED) estimates of high precision.

The placement of doses and allocation of experimental subjects to doses can substantially affect the precision with which an ED is estimated (Smith and others [in press]). The problem of determining the optimal design for a fixed total sample size has been addressed to some extent in the statistical literature (e.g., Abdelbasit and Plackett 1983, Brown 1966, Freeman 1970, Tsutakawa 1980); but the most readily accessible guidelines for bioassay design are probably those by Finney (1971). Finney's guidelines are applicable for estimating ED's in the vicinity of the 50 percent response level.

Research with chemicals often requires precise estimation of an ED at the extremes of the response curve, rather than in the middle. For example, derivation of multiplication factors to predict mortality rates in the field on the basis of laboratory data necessitates estimating the ED_{90} or ED_{95} (Haverty and Robertson 1982). Assessment of the toxicity of an insecticide to an endangered or threatened species, on the other hand, entails estimation of ED's at the low extremes of the response curve, such as the ED_{10}. Although it is intuitively obvious that the optimal designs for estimating the ED_{10} and ED_{90} will differ from each other and from the optimal design for the ED_{50}, guidelines for designing experiments expressly for the purpose of estimating an extreme quantile are generally lacking.

DOSESCREEN is a computer program written to assist investigators in dose placement. It is a computationally simple, yet flexible, tool with which to plan experiments that involve a binary quantal response model with one independent variable. A measure of precision for an estimator of an ED is produced on the basis of the asymptotic expected length of the confidence interval for the ED with a proposed experimental design. The measure represents a generalization of the measure derived by Finney (1971); DOSESCREEN can provide an estimate of precision for any ED calculated with any of a large class of tolerance distributions.

On the basis of the DOSESCREEN measure of precision, an efficient design can be selected for estimating an ED of interest. If two or more ED's are to be estimated from the same experiment, DOSESCREEN can help the investigator find a good compromise design, i.e., one that results in reasonably high precision for estimates of all the ED's of interest. DOSESCREEN output may also provide assistance in selecting the total sample size for an experiment.

Although the language adopted in this report is oriented to entomological bioassays, DOSESCREEN can be used to design experiments in other scientific disciplines in which the levels of the independent variable are controlled by the experimenter. In agricultural research, for example, nursery studies of the effect of fertilizers or of water stress on seedling survival might be improved by careful consideration of treatment level placement. Clinical trials in medicine provide another example, and it is easy to conceive of analogous experimental situations in the behavioral sciences.

This report describes the statistical basis of the DOSESCREEN measure of precision, suggests uses of it, and presents a hypothetical experiment designed on the basis of program output. The DOSESCREEN computer program listed in the *appendix is* written in Fortran 77 so that users can easily convert DOSESCREEN to their own computer system.

1. STATISTICAL BASIS FOR DOSESCREEN

1.1 Binary Quantal Response Models

A dose-response bioassay typically involves selecting T dose levels of a chemical, administering the t^{th} dose level x_t to n_t test subjects, and denoting the response of each subject as either 1 (for example, dead) or 0 (for example, alive). The numbers of subjects per dose level, n_t, are called the *cell sizes* for the experiment. The total number of subjects, $N = \Sigma n_t$, is called the *sample size.*

The statistical framework for this type of experiment is the binary quantal response model with one independent variable. Subjects that receive the same dose are assumed to have the same probability of responding, and probability of response is assumed to be functionally related to the dose level in the form

$$P_t = F(\beta_1 + \beta_2 x_t)$$

where P_t is the probability of response, x_t is the dose level, and $F(\cdot)$ is a cumulative distribution function (CDF) with density $f(\cdot)$ and inverse $F^{-1}(\cdot)$. The CDF's most commonly used by biologists are the Gaussian

$$F(x) = \int_{-\infty}^{x} e^{-z^2} / \sqrt{2\pi} dz$$

which results in a model traditionally called the probit model, and the logistic

$$F(x) = 1 / (1 + e\text{-}x)$$

which results in the logit model.

A number of procedures that provide estimators b_1, b_2 of the regression coefficients β_1, β_2 are available. Those most widely used belong to the class of regular best asymptotically normal (RBAN) estimators; these estimators have sampling distributions that approach the same bivariate normal distribution as all the cell sizes approach infinity. For example, the maximum likelihood and minimum logit chi-square estimators are both RBAN estimators. For large n_t,

$$b_1, b_2 \sim N(\beta_1, \beta_2, v_{11}, v_{22}, v_{12}$$

where

$$\begin{bmatrix} v_{11} & v_{12} \\ v_{12} & v_{22} \end{bmatrix} = \begin{bmatrix} \Sigma n_t w_t & \Sigma n_t w_t x_t \\ \Sigma n_t w_t x_t & \Sigma n_t w_t x_t^2 \end{bmatrix}^{-1} \quad (1)$$

with $w_t = [f(\beta_1 + \beta_2 x_t)]^2 / P_t(1-P_t)$. Classical inference is based on the use of an RBAN estimator with cell sizes large enough to ensure that the normal distribution provides a good approximation to the distribution of b_1, b_2.

1.2 Effective Doses (ED's) and Confidence Intervals

Let μ_0 denote the dose level that produces a certain probability of response P_0, where $0 < P_0 < 1$:

$$\mu_0 = [F^{-1}(P_0) - \beta_1] / \beta_2.$$

For example, if $P_0 = 0.5$, then μ_0 is the ED_{50}. An estimate m_0 of μ_{10} is obtained by substituting b_1 and b_2 for, β_1 and β_2 in the expression for μ_0,

$$m_0, = [F^{-1}(P_0) - b_1] / b_2.$$

For large samples, the $(1-\alpha)$ 100 percent confidence limits for μ_0 (Finney 1971) are

$$\frac{m_0 + g(m_0 + v_{12}/v_{22}) \pm z\sqrt{v_{11} + 2m_0 v_{12} + m_0^2 v_{22} - g(v_{11} - v_{12}^2/v_{22})}}{1-g} \Big/ b_2(1-g)$$

where $v_{ij}, i,j = 1,2$ are the estimated variances and covariance of b_1, b_2 computed by substitution of the estimated parameters for their true values in (1), z is the $(1-\alpha/2)100$ percent quantile of the normal (0,1) distribution, and $g = z^2 v_{22} / b_2^2$.

1.3 DOSESCREEN Measure of Precision

The length of the confidence interval for μ_0 is

$$L = 2z\sqrt{v_{11} + 2m_0 v_{12} + m_0^2 v_{22} - g(v_{11} - v_{12}^2/v_{22})} \Big/ b_2(1-g) \quad \text{if } g<1 \quad (2)$$

$$= \infty \quad \text{if } g \geq 1.$$

For any finite sample size, the expected ("average") value of L is infinite, because L is infinite with positive probability; however, as all cell sizes approach infinity, the probability that $g \geq 1$ approaches zero if $\beta_2 \neq 0$, so that asymptotically L has finite expectation E(L). A first order approximation to E(L), denoted by $L^*\beta_2$, is found by replacing all the random variables in (2) by their asymptotic expected values; e.g., b_i; is replaced by β_i , v_{ij} by v_{ij} and m_0 by μ_0.

$L^*\beta_2$ appears to be a complicated function of the unknown parameters and the dose levels. However, simple algebra shows that L* is actually a function only of the underlying probabilities of response P_t. For this reason, L* rather than $L^*\beta_2$ was chosen as the DOSESCREEN measure of precision:

$$L^* = 2z\sqrt{v_{11}^* + 2F^{-1}(P_0)v_{12}^* + [F^{-1}(P_0)]^2 v_{22}^* - g^*(v_{11}^* - v_{12}^{*2}/v_{22}^*)} \Big/ 1-g^*$$

where

$$\begin{bmatrix} v_{11}^* & v_{12}^* \\ v_{12}^* & v_{22}^* \end{bmatrix} = \begin{bmatrix} \Sigma n_t w_t^* & \Sigma n_t w_t^* F^{-1}(P_t) \\ \Sigma n_t w_t^* F^{-1}(P_t) & \Sigma n_t w_t^* [F^{-1}(P_t)]^2 \end{bmatrix}^{-1}$$

with $w_t^* = f(F^{-1}(P_t))2/P_t(1-P_t)$, and $g^* = z^2 v_{22}$.*

When applied to the ED_{50}, L* is related to Finney's (1971, p. 142) measure of precision, Ib^2N, by $Ib^2N = (\Sigma n_t)L^{*2} / 4$. L* is not an approximation to confidence interval length unless $\beta_2 = 1$, but is a measure of precision suitable for evaluating the relative performance to be expected from candidate designs for the same experiment, because for any one experiment β_2 will be a constant, irrespective of the design.

1.4 Factors Affecting Accuracy of Approximation

The approximation L* predicts the relative precision of an experimental design over repeated identical experiments. The result of any one experiment will unavoidably depart from the prediction because of random variation alone. However, other factors which are largely under the control of the investigator can introduce error into L*. These are sample size, choice of model, and dose selection.

L* is based on large sample approximations to the sampling distributions of b_1, and b_2 and cannot be expected to be reliable for small samples. A Monte Carlo experiment was performed to evaluate the accuracy of L* using a logit model with $\beta_1 = 0$, $\beta_2 = 1$, and selected designs (Smith and others [in press]). The average and median lengths of confidence intervals produced in 3000 replications using the maximum likelihood estimator were compared with those predicted by L* (*table 1*). The results indicated that L* underestimates the average length in small samples, but is a highly accurate estimator of median interval length.

The recommended placement of doses will depend on the underlying model assumed, that is, on the choice of F. Use of an inappropriate model will introduce bias in L*. Our experience has been that optimal designs for the probit and logit models are virtually identical, but different models generally result in different optimal designs, particularly for estimating an extreme ED.

Accuracy of the approximation also depends on an accurate determination of the dose levels x_t corresponding to the P_t selected. In practice some error will always be introduced in moving from P_t to x_t (Abdelbasit and Plackett 1983). The error can be minimized by conducting preliminary experiments to establish a tentative dose-response relationship for the chemical under investigation.

2. SUGGESTED USES OF DOSESCREEN

DOSESCREEN computes a measure of precision L* for any combination of cell sizes (n_t), response probabilities (P_t), ED, and desired significance level (α). The program is particularly well-suited for use interactively; in this mode, an investigator can explore how precision is affected by sample size, number of dose levels and dose placement, and allocation of sample size to dose levels. DOSESCREEN may be used in other ways as well.

Table 1--*DOSESCREEN (L*) predicted average 95 percent confidence interval length and average and median lengths obtained from Monte Carlo (M. C.) simulation with a logit model*[1]

Response probabilities	Cell size	ED_{50}					ED_{90}				
		L*	M.C.[2] avg.	(Error)	median M.C.	(Error)	L*	M.C. avg.	(Error)	M.C. median	(Error)
				pct		*pct*			*pct*		*pct*
0.20,0.35,0.50,0.65,0.80	24	0.885	0.968	(-8.2)	0.888	(-0.3)	2.635	3.272	(-20)	2.591	(1.7)
	48	.590	.605	(-2.6)	.590	(-.1)	1.665	1.763	(-5.6)	1.647	(1.1)
	96	.406	.411	(-1.3)	.406	(-.2)	1.118	1.148	(-2.6)	1.113	(.5)
0.10,0.80,0.85,0.90,0.95	24	1.492	1.547	(-3.6)	1.497	(-.3)	1.461	1.494	(-2.2)	1.459	(.1)
	48	1.008	1.016	(-.8)	1.006	(.2)	.988	.992	(-.4)	.989	(-.2)
	96	.698	.700	(-.4)	.696	(.2)	.684	.687	(-.4)	.685	(-.1)
0.20,0.30,0.40,0.45	15	.894	1.050	(-15)	.899	(-.6)	3.084	4.574	(-33)	2.997	(2.9)
0.55,0.60,0.70,0.80	30	.585	.609	(-4.0)	.587	(-.4)	1.879	2.052	(-8.4)	1.859	(1.1)
	60	.399	.405	(-1.6)	.400	(-.2)	1.244	1.287	(-3.4)	1.236	(.6)
0.10,0.15,0.70,0.75	15	1.196	1.207	(-1.2)	1.182	(.9)	1.557	1.601	(-2.8)	1.552	(.3)
0.80,0.85,0.90,0.95	30	.811	.813	(-.2)	.806	(.6)	1.048	1.056	(-.7)	1.038	(1.0)
	60	.563	.564	(-.2)	.561	(A)	.724	.728	(-.6)	.722	(.3)

[1]Regression coefficients, $\beta_1 = 0$, $\beta_2 = 1$, were fixed.

[2]Average length of confidence interval excluded samples that resulted in infinite confidence intervals. Results are for the maximum likelihood estimator in 3000 replications

2.1 Determining Optimal Design

The DOSESCREEN subroutine can be easily inserted into a computer routine that computes L* for all possible values of P_t, n_t: $\Sigma n_t = N$ to find the optimal placement of doses for a given model, fixed total sample size (N), number of doses (T), and ED. Because of computer time constraints, in our computer routine we restricted the search to values of P_t that are multiples of 0.05 and required equal numbers of subjects per dose, so that $n_t = N/T$. Optimal designs subject to these constraints are reported in *table 2* for estimation of the ED_{50} and ED_{90}, assuming a logit model. Because of symmetry, the optimal design for estimation of the ED_{10} can be found from that of the ED_{90} by translation of the P_t. For example, optimal P_t for the ED_{10} at N = 240, T = 3 is (0.15, 0.20, 0.95).

The recommended designs do not change much as N increases. For estimating the ED_{50}, doses should be placed symmetrically on the response curve about the ED_{50}. For increasing N, the doses should be placed somewhat closer together. For precise estimates of the ED_{90}, the majority of the doses should be located in the vicinity of the ED_{90}, with one or two doses located in the region of low response. On the basis of L*, confidence interval length apparently is relatively insensitive to the number of dose levels chosen.

2.2 Finding Compromise Designs

Often an investigator wishes to estimate simultaneously more than one ED from one experiment. For example, in insecticide research, the ED_{90} is often of primary interest, but in the literature the ED_{50} estimate is always reported as well. Since dramatically different designs are recommended for the two ED's, a third design may be desired that will result in moderately high precision for both the ED_{50} and the ED_{90}.

The global search for the optimal design can be modified easily to provide the information necessary to select a good compromise design. Our search program divides all possible designs into categories with similar values of L* for the ED_{50}. Designs in each category are searched separately to find the optimal design for the ED_{90}.

An example of this technique is presented in *table 3*. Because $NL^{*2}/4$ remains relatively stable and of moderate size as N varies, $NL^{*2}/4$ was computed instead of L*. In the example in *table 3* and in all others we examined, the optimal ED_{50} design poorly estimated the ED_{90}, whereas the optimal design for the ED_{90} estimated the ED_{50} rather

Table 3--*Optimal designs for estimating the ED_{90}, by categories of the ED_{50}* [1]

Categories of $NL^{*2}/4$ for the ED_{50}[2]	Optimal design (P_t) within category for the ED_{90}	Value of $NL^{*2}/4$ for optimal design	
		ED_{50}	ED_{90}
Optimal for ED_{50}	0.30,0.35,0.40,0.70,0.75	20.36	213.06
20-30	.15, .30, .80, .85, .90	29.84	74.98
30-40	.15, .65, .80, .85, .90	39.25	67.04
40-50	.15, .75, .85, .90, .95	49.59	61.18
>50	.10, .80, .85, .90, .95	60.98	58.36

[1]Based on a logic model with five doses, a total sample size of N = 240, equal cell sizes, and a level of significance. α= 0.05. Designs were considered for which all P_t were multiples of 0.05

[2]L* = DOSESCREEN measure of precision.

Table 2--*Optimal designs for estimating the ED_{50} and ED_{90} for a logit model*[1]

Sample size	No. doses	Cell size	ED_{50} Optimal P_t	ED_{50} L*	ED_{90} Optimal P_t	ED_{90} L*
120	3	40	0.25, 0.50, 0.75	0.88	0.05,0.85,0.90	1.51
	5	24	.25, .30, .50, 0.70, 0.75	.88	.10, .80, .85,0.90,0.95	1.46
	6	20	.25, .30, .35, .65, .70,0.75	.88	.10, .75, .80, .85, .90,0.95	1.49
	8	15	.20, .25, .35, .40	.88	.05, .10, .70, .75	1.54
			.60, .65, .75, .80		.80, .85, .90, .95	
240	3	80	.30, .45, .75	.58	.05, .80, .85	1.02
	5	48	.30, .35, .45, .70, .75	.58	.10, .80, .85, .90, .95	.99
	6	40	.25, .35, .40, .60, .65, .75	.58	.10, .75, .80, .85, .90, .95	.99
	8	30	.25, .30, .35, .40	.58	.05, .10, .70, .75	1.04
			.60, .65, .70, .75		.80, .85, .90, .95	
480	3	160	.30, .55, .65	.40	.05, .80, .85	.71
	5	96	.30, .40, .45, .65, .70	.39	.05, .80, .85, .90, .95	.68
	6	80	.30, .35, .40, .60, .65, .70	.39	.05, .75, .80, .85, .90, .95	.68
	8	60	.30, .35, .40, .45	.39	.05, .10, .70, .75	.72
			.50, .60, .70, .75		.80, .85, .90, .95	
720	3	240	.35, .50, .65	.32	.05, .80, .85	.57
	5	144	.30, .40, .55, .60, .65	.32	.05, .80, .85, .90, .95	.55
	6	120	.30, .40, .45 .50, .65, .70	.32	.05, .75, .80, .85, .90, .95	.55
	8	90	.30, .35, .40, .45	.32	.05, .10, .70, .75	.58
			.55, .60, .65, .70		.80, .85, .90, .95	

[1]P_t = probability of response

L* = DOSESCREEN measure of precision for optimal design.

well. *Table 3* presents designs that estimate the ED_{90} with progressively greater precision, at the expense of the precision of the ED_{50} estimate. With this tool, a compromise design can be chosen intelligently.

2.3 Determining Sample Size

Another possible application of DOSESCREEN is determining a sample size large enough so that the experiment will produce an ED confidence interval of a certain median length L_0. This application requires a prior estimate of β_2. If $\tilde{\beta}_2$ is an estimate of β_2, DOSESCREEN can be put into a loop to determine N such that $L^* = L_0 \tilde{\beta}_2$, for fixed P_t and n_t / N, the proportional allocation of subjects to dose levels.

When using DOSESCREEN to determine sample size, L_0 must be specified in the same units as x_t. If, for example, x_t is in units of log-concentration--as is often the case in chemical bioassays--then L_0 must represent the desired length for a confidence interval for the log ED. This requirement limits the usefulness of DOSESCREEN for sample size determination because no simple relationship exists between confidence interval length for the log ED and the corresponding confidence interval length for the ED in the original units of concentration.

DESIGNING AN EXPERIMENT WITH DOSESCREEEN

To demonstrate how DOSESCREEN can be used to plan a chemical bioassay, we present a hypothetical experiment designed on the basis of DOSESCREEN output.

An experiment is being planned to investigate the toxicity of malathion to tent caterpillars. On the basis of a concentration-bracketing experiment, it appears reasonable to assume a probit model using a $\log_{10}$ transformation of the concentrations, which are in units of parts per million (ppm); intercept and slope estimates from the initial experiment were $\tilde{\beta}_1 = 1.14$, $\tilde{\beta}_2 = 1.01$. The planned experiment is primarily for the purpose of estimating the ED_{90}, but 95 percent confidence intervals for the ED_{50} and the ED_{95} will be reported as well. The investigator has 180 insects available for the study, and intends to use three to nine dose levels.

The following questions are explored by using DOSESCREEN interactively:

- For precise estimation of the ED_{90}, where on the interval (0,1) should the response probabilities lie?
- What is the effect of varying the number of dose levels?
- Should all doses have the same number of subjects, or can one improve precision by unequal allocation of subjects to doses?
- How well can the ED_{95} and ED_{50} be estimated with a design selected specifically to estimate the ED_{90}?

Using a design with five doses, a total sample of N = 180, and equal cell sizes, three possible distributions of response probabilities P_t, t = 1,5, are considered: an even distribution between 0.10 and 0.90; placement of all the response probabilities on the upper half of the response curve; and an intermediate design, in which P, is on the low end of the response curve, the other P_t located on the upper portion.

P_t					L* for ED_{90}
0.10	0.30	0.50	0.70	0.90	0.87
.50	.60	.70	.80	.90	1.09
.10	.60	.70	.80	.90	.71

On this evidence, a design that places most of the doses near the ED_{90} and a small number of P_t on the lower portion of the response curve is desirable.

The effect of varying the number of dose levels is next explored by considering the use of three, five, or nine levels, again with N = 180 and equal cell sizes:

P_t									L* for ED_{90}
0.05	0.90	0.95							0.66
.05	.80	.85	0.90	0.95					.59
.05	.10	.15	.70	.75	0.80	0.85	0.90	0.95	.67

Five doses appear preferable to either three or nine doses.

There are numerous ways to allocate the 180 subjects to the five doses. Here, we consider three: equal allocation ($n_t = 36$), assignment of more subjects to P_t at the upper and lower extremes ($n_1 = n_5 = 54$, $n_2 = n_3 = n_4 = 24$), or assignment of more subjects to intermediate values of P, ($n_, = n_5 = 18$, $n_2 = n_3 = n_4 = 48$). When these allocation schemes are used with the best design encountered thus far, the following values of L* are obtained:

P_t					
0.05	0.80	0.85	0.90	0.95	L* for ED_{90}
36	36	36	36	36	0.59
54	24	24	24	54	.63
18	48	48	48	18	.59

For estimation of the ED_{90}, two of the three allocation schemes appear to be equally precise. To choose between them, we examine their relative precision for estimating the ED_{95} and the ED_{50} at N = 180:

P_t						
0.05	0.80	0.85	0.90	0.95	L* for ED_{95}	L* for ED50
36	36	36	36	36	0.71	0.66
18	48	48	48	18	.78	.88

Clearly, the equal allocation results in higher precision for both the ED_{95} and ED_{50}, and is thus preferable to the allocation scheme that assigns a higher proportion of subjects to the central values of P_t.

At this point, we have narrowed the field of possible experimental designs to those with equal allocation of subjects to doses. Further, the doses will be placed so that one dose results in a low probability of response, with the others located so as to produce response probabilities between 80 and 95 percent.

The final step entails selection of doses that will faithfully reproduce the desired response probabilities. This is done by using the estimates of the regression coefficients and the identity

$$P_t = \Phi(\tilde{\beta}_1 + \tilde{\beta}_2 x_t)$$

or

$$x_t = [\phi^{-1}(P_t) - \tilde{\beta}_1]/\tilde{\beta}_2$$

where $\Phi(\cdot)$ represents the standard normal cumulative distribution function and $\Phi^{-1}(\cdot)$ is its inverse. A computer program that computes $\Phi^{-1}(\cdot)$ or a standard normal table is required to solve for x_t. For our hypothetical experiment, the doses in the original units of concentration would be $10^x t$:

P_t				
0.05	0.80	0.85	0.90	0.95
ppm				
0.0017	0.51	0.79	1.39	3.13

4. APPENDIX

4.1 DOSESCREEN Subroutine and Auxiliary Subroutines

```
      SUBROUTINE DOSESC(Z,NDOSES,N,P,P0,W,FIP,LSTAR,IER)
C
C ***************************************************************************
C This subroutine computes LSTAR, a measure of precision for the estimation
C of an ED in a quantal response model with one independent variable.
C
C Argument list:
C       Input:  Z     : the (1-ALPHA/2)*100 percent quantile of the
C                       normal(0,1) distribution.
C
C               NDOSES: number of dose levels specified.
C
C               N(I)  : number of subjects at dose level I, I=1,NDOSES
C
C               P(I)  : probability of response at dose level I, I=1,NDOSES
C
C               P0    : the probability of response associated with the
C                       ED of interest. For example, if interest lies in
C                       the ED90, then P0 = 0.9.
C
C       Output: W(I)  : the weight associated with dose level I, I=1,DOSES
C
C               FIP(I): F inverse of P(I), computed by subroutine FUNCT,
C                       I=1,NDOSES
C
C               LSTAR : measure of precision.
C
C               IER   : error code
C                       IER = 1 if successful completion of computation
C                           = 2 if G is greater than or equal to 1.0,
C                             indicating that LSTAR is infinite.
C                           = 3 if input P(I) or P0 is not in (0,1).
C                           = 4 if N(I) is not greater than 0.
C                           = 5 if SIGMA or SIGMA inverse is nearly
C                             singular to within a tolerance of 1.0*E-8.
C
C
C subroutines called by DOSESC:
C
C       CHKPT:  verifies that P(I) is in (0,1) and that N(I) is at least 1.
C
C       FUNCT:  user supplied subroutine which computes F inverse of P0,
C               of P(I) and weights W(I) corresponding to user's choice
C               of model.
C
C ***************************************************************************
```

```
      DIMENSION P(NDOSES),N(NDOSES),W(NDOSES),FIP(NDOSES),SIGMA(2,2)
      REAL LSTAR
C
      CALL CHKPT(NDOSES,N,P,P0,IER)
      IF (IER.NE.1) RETURN
C
      CALL FUNCT(NDOSES,P,P0,FIP,FIP0,W)
C
C Compute SIGMA, the covariance matrix of the parameter estimates.
C
      SNW=0.
      SNWX=0.
      SNWXX=0.
C
      DO 100 I=1,NDOSES
            SNW=SNW + N(I)*W(I)
            SNWX = SNWX + N(I)*W(I)*FIP(I)
            SNWXX = SNWXX + N(I)*W(I)*FIP(I)*FIP(I)
  100 CONTINUE
C
C Check on singularity of SIGMA. Determinant of SIGMA is 1/DET.
C
      DET = SNW*SNWXX - SNWX*SNWX
      IF (DET.GT.1.E-8.AND.DET.LT.1.E8) GO TO 200
C
C Sigma will be nearly singular. Set IER to 5 and RETURN
C
      IER = 5
      RETURN
C
  200 SIGMA(1,1) = SNWXX/DET
      SIGMA(1,2) = -SNWX/DET
      SIGMA(2,2) = SNW/DET
C
C Compute G and verify that G is in interval (0,1)
C
      G=Z*Z*SIGMA(2,2)
      IF (G.GE.0.0.AND.G.LT.1.0) GO TO 300
C
C G is not in the interval (0,1).Set IER to 2 and RETURN.
C
      IER = 2
      RETURN
C
  300 A = 2.*Z/(1.-G)
      B = SIGMA(1,1) + 2.*SIGMA(1,2)*FIP0 + SIGMA(2,2)*FIP0*FIP0
      C = G*(SIGMA(1,1) - SIGMA(1,2)**2/SIGMA(2,2))
C
      LSTAR = A * SQRT(B-C)
      RETURN
      END
```

```
      SUBROUTINE CHKPT(NDOSES,N,P,P0,IER)
C
C ***************************************************************************
C This subroutine verifies that P(I) and P0 all lie in (0,1) and that
C N(I) is greater than or equal to 1, I=1,NDOSES
C
C ***************************************************************************
      DIMENSION N(NDOSES),P(NDOSES)
      IER = 1
C
C check on P0
C
      IF (P0.GT.0.0.AND.P0.LT.1.0) GO TO 100
C
C o.w. IER equals 3 and RETURN
C
      IER = 3
      RETURN
C
  100 DO 300 I=1,NDOSES
            IF (P(I).GT.0.0.AND.P(I).LT.1.0) GO TO 200
C
C o.w. IER equals 3 and RETURN
            IER = 3
            RETURN
C
  200       IF (N(I).GE.1) GO TO 300
C
C o.w. IER equals 4 and RETURN
C
            IER = 4
            RETURN
  300 CONTINUE
C
      RETURN
      END
```

```
      SUBROUTINE FUNCT(NDOSES,P,P0,FIP,FIP0,W)
C
C ***************************************************************************
C This is a sample of a subroutine FUNCT which is appropriate when a
C Logit model is assumed; that is
C
C              P = F(X) = 1/(1 + EXP(-X))
C
C with density function
C
C              F'(X) = EXP(-X)/(1 + EXP(-X))**2
C
C                    = F(X)*(1-F(X))
C
C and inverse function
C
C              FI(P) = LOG(P/(1-P)).
C
C
C
C
C Therefore    FIP0 = LOG(P0/(1-P0)),
C
C              FIP(I) = LOG(P(I)/(1-P(I))), I=1,NDOSES
C and
C              W(I) = F'(FIP(I))**2/(P(I)*(1-P(I)))
C
C                   = P(I)*(1-P(I)), I=1,NDOSES.
C
C ***************************************************************************
      DIMENSION P(NDOSES),FIP(NDOSES),W(NDOSES)
C
      FIP0 = ALOG(P0/(1.-P0))
C
      DO 100 I=1,NDOSES
            FIP(I) = ALOG(P(I)/(1.-P(I)))
            W(I) = P(I)*(1.-P(I))
  100 CONTINUE
C
      RETURN
      END
```

4.2 Sample Main Program and Output

```
C ***************************************************************************
C This is a sample main routine which calls the DOSESC subroutine. In this
C example, LSTAR is computed for the ED50; three dose levels are proposed,
C with 50 subjects assigned to each dose level. Response probabilities
C are 0.20,0.50, and 0.80.
c
C The value of Z associated with ALPHA = 0.05 is input directly here.
C To increase the accuracy of Z, one may supply a subroutine that
C numerically evaluates the inverse gaussian distribution function at
C the value (1-ALPHA/2).
C
C ***************************************************************************
      DIMENSION P(3),N(3),W(3),FIP(3)
      REAL LSTAR
      DATA ALPHA/.05/,Z/1.95996/,N/3*50/,P/.2,.5,.8/,P0/.5/,NDOSES/3/,
     1 NIO/6/
C
      CALL DOSESC(Z,NDOSES,N,P,P0,W,FIP,LSTAR,IER)
C
      WRITE (NIO,20)
      DO 10  I=1,NDOSES
             WRITE(NIO,30) N(I), P(I)
 10   CONTINUE
 20   FORMAT (10X,"DOSESCREEN ESTIMATE LSTAR"/
     1 "            N            P          ")
 30   FORMAT (10X,I4,10X,F4.3)
C
      GO TO (100,200,300,400,500), IER
C
 100  NP0 = 100*P0
      NSIGN = 100*(1.-ALPHA)
      WRITE(NIO,101) LSTAR,NSIGN,NP0
      STOP
 101  FORMAT (10X,"LSTAR = ",F10.4," FOR A ",I2,"% CI FOR THE ED",i2)
C
 200  WRITE(NIO,201)
      STOP
 201  FORMAT(10X,"ESTIMATED CONFIDENCE INTERVAL FOR THIS DATA INFINITE")
C
 300  WRITE(NIO,301) P0
      STOP
 301  FORMAT(10X,"P(I) OR P0 = ",F7.3," NOT IN THE INTERVAL (0,1)")
C
 400  WRITE(NIO,401)
      STOP
 401  FORMAT(10X,"N(I) NOT AN INTEGER GREATER THAN OR EQUAL TO 1")
C
 500  WRITE(NIO,501)
      STOP
 501  FORMAT(10X,"AN ALMOST SINGULAR COVARIANCE MATRIX DETECTED")
C
      END
```

Output from sample Main program

```
DOSESCREEN ESTIMATE LSTAR
  N            P
  50          .200
  50          .500
  50          .800
LSTAR =     .0641 FOR A 95% CI FOR THE ED50
```

5. REFERENCES

Abdelbasit, K. M.; Plackett, R. L. **Experimental design for binary data.** J. Am. Stat. Assoc. 78:90-98; 1983.

Brown, B. W., Jr. **Planning a quantal assay of potency.** Biometrics 22:322-329; 1966.

Finney, D. J. **Probit analysis.** 3d ed. London: Cambridge University Press; 1971. 333 p.

Freeman, P. R. **Optimal Bayesian sequential estimation of the median effective dose.** Biometrika 57:79-89; 1970.

Haverty, Michael I.; Robertson, Jacqueline L. **Laboratory bioassays for selecting candidate insecticides and application rates for field tests on the western spruce budworm**. J. Econ. Entomol. 75(2):179-182; 1982 April.

Smith, Kimberly C.; Savin, N. E.; Robertson, Jacqueline L. **A Monte Carlo comparison of maximum likelihood and minimum chi-square sampling distributions in logit analysis.** Biometrics (In press).

Tsutakawa, Robert K. **Selection of dose levels for estimating a percentage point of a logistic quantal response curve.** Appl. Stat. 29:25-33; 1980.

Smith, Kimberly C.; Robertson, Jacqueline L. **DOSESCREEN: a computer program to aid dose placement.** Gen. Tech. Rep. PS W-78. Berkeley, CA: Pacific Southwest Forest and Range Experiment Station, Forest Service, U.S. Department of Agriculture; 1984. 12 p.

Careful selection of an experimental design for a bioassay substantially improves the precision of effective dose (ED) estimates. Design considerations typically include determination of sample size, dose selection, and allocation of subjects to doses. DOSESCREEN is a computer program written to help investigators select an efficient design for the estimation of an arbitrary ED. This report establishes the statistical basis for DOSESCREEN and suggests several ways to utilize DOSESCREEN output. A copy of the computer program in Fortran 77 is provided so users can easily convert DOSESCREEN to their own computer system.

Retrieval terms: bioassay, experimental design, effective doses